EPRI | ELECTRIC POWER RESEARCH INSTITUTE

OpenDSS Training Workshop

PVSystem, InvControl, Storage, StorageController – Parts I and III

Celso Rocha, Paulo Radatz
EPRI Knoxville, TN

August 26, 2020

Instructor

▪ Celso Rocha, Member, *IEEE*

Celso Rocha serves as Engineer Scientist II at the Electric Power Research Institute (EPRI) in Knoxville, Tennessee USA. He received his degree in electrical engineering with emphasis in energy and automation from University of Sao Paulo, Sao Paulo, Brazil (2017); Currently, he is pursuing his Masters in electrical engineering at University of Sao Paulo, Sao Paulo, Brazil under the supervision of Dr. Nelson Kagan. He has 5 years of experience with OpenDSS, having taught several OpenDSS trainings in Brazil at conferences, universities and industry.

▪ Paulo Radatz

Paulo Radatz serves as Engineer Scientist II at the Electric Power Research Institute (EPRI) in Knoxville, Tennessee USA. He received both his Masters and Bachelors degree in electrical engineering with emphasis in energy and automation from University of Sao Paulo, Sao Paulo, Brazil. He was awarded a prize for being the best bachelor's student of Polytechnic School of University of Sao Paulo (2015). He has 5 years of experience with OpenDSS, having taught several OpenDSS trainings in Brazil at conferences, universities and industry.

1. Inverter Modeling in OpenDSS

“New inverter.*”?

- No standalone inverter component
- Features span over PVSystem, Storage and InvControl
  - At PVSystem and Storage:
    - Nameplate, settings and losses: kVA, kvarMax, …
    - Basic functions: Constant PF, constant kvar, …
  - At InvControl:
    - More complex functionalities (Volt-Var, DRC, Volt-Watt, …)



Inverter Capability Curve



Inverter Capability Curve

- Customization: set the respective parameter to “0” to disable it



Inverter Capability Curve

- Violation of capability curve: application of priorities
  - *wattPriority* [True/False*]
  - *pfPriority* [True/False*]



2. PVSystem + InvControl

PVSystem Element in OpenDSS

- The PVSystem model combines the photovoltaic (PV) array and the PV inverter into one convenient model to use for distribution system impacts studies



PVSystem Properties

PV Array

Pmpp

P-TCurve

PV Inverter

Inverter Connection and Efficiency Curve

kV, Phases, bus1, conn, EffCurve

Inverter Capability Curve

%cutin, %cutout, kvarMax, kvarMaxAbs, WattPriority, PFPriority, %PminNoVars, %PminkvarMax

Inverter Functions

%Pmpp, PF, kvar, VarFollowInverter

Operating Conditions

SnapShot

Irradiance, Temperature

QSTS

Irradiance,

irradiance curve (daily, yearly, or duty),

temperature curve (Tdaily, Tyearly, or Tduty)

Example Script






```
Clear

New Circuit.TheveninEquivalent bus1=A pu=1.0 basekv=13.8

New XYCurve.Eff npts=4 xarray=[.1  .2  .4  1.0]  yarray=[.86  .9  .93  .97]
New XYCurve.FatorPvsT npts=4 xarray=[0  25  75  100]  yarray=[1.2 1.0 0.8  0.6]

New Loadshape.Irrad npts=24 interval=1
~ mult=[0 0 0 0 0 0 .1 .2 .3 .5 .8 .9 1.0 1.0 .99 .9 .7 .4 .1 0 0 0 0 0]

New Tshape.Temp npts=24 interval=1
~ temp=[25 25 25 25 25 25 25 25 35 40 45 50 60 60 55 40 35 30 25 25 25 25 25 25]

New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1200 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp pf=0.9

Set voltagebases=[13.8]
Calcvoltagebases
Set mode=daily
Solve
```

Base Case

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1200 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp pf=0.9 %cutin=20 %cutout=20
```



Pdc < %cutin * kVA

Pdc < %cutout * kVA

Base Case

```
New Monitor.PV_v element=PVSystem.PV terminal=1 mode=3
```



kVA exceeded under P priority

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1000 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp wattpriority=yes pf=0.8
```



kVA exceeded under Q priority

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1000 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp wattpriority=no pf=0.8
```



Active Power Limited – Limit DER Power

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1000 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=60 daily=Irrad Tdaily=Temp pf=1
```

Active power limited to *%Pmpp* x *Pmpp*

Curtailment



Reactive Power Limited

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1000 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp pf=0.6 kvarmax=300
```



Night Operation

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1000 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp kvar=300 varfollowinverter=no
```



InvControl Element in OpenDSS

- Works in conjunction with PVSystem/Storage object(s) to control the DERs output according to advanced ('smart') inverter functions



InvControl Properties

Common Properties

Mode and element controlled

DERList, mode, Combimode

Convergence

VoltageChangeTolerance, VarChangeTolerance, ActivePChangeTolerance

deltaQ_factor, deltaP_factor (Set equal to -1 allows automatic algorithm)

Monitored Voltage

monBus, monBusesVbase, monVoltageCalc, voltage_curvex_ref

REQUEST Reactive Power

RefReactivePower

Volt-var

Vvc_curve1

DRC (Dynamic Reactive Current)

DbvMin, DbvMax, ArGraLowV, ArGraHiV, DynReacavgwindowlen

Watt-pf

Wattpf_curve

Watt-var

Wattvar_curve

LIMIT Active Power

Volt-watt

VoltwattYAxis

Voltwatt_curve

VoltwattCH_curve (Storage Only)

Smart Inverter Functions

- Volt-Var
  - Follows a voltage versus reactive power curve and **REQUESTS** the reactive power generation (capacitive) or reactive power absorption (inductive) according to the monitored voltage of each DER element



Smart Inverter Functions

- Volt-Watt
  - Follows a voltage versus active power curve and defines the active power output **LIMIT** according to the monitored voltage of each DER element



Smart Inverter Functions

- Watt-PF
  - Follows an active power versus power factor curve to **REQUEST** the reactive power according to the active power output of each DER element
- Watt-Var
  - Follows an active power versus reactive power curve and **REQUESTS** the reactive power according to the active power output of each DER element





Smart Inverter Functions

- Dynamic Reactive Current (DRC)
  - Has several settings that **REQUEST** the reactive power generation or absorption in response to fast changes in monitored voltage (e.g., during a sag or swell)
- Volt-Var + Volt-Watt
- Volt-Var + DRC
- Volt-Watt + DRC
- Volt-Watt + PF or var constant
  - Volt-Watt set in InvControl and PF or var constant set in PVSystem/Storage

Volt-Var Example

```
New PVSystem.PV phases=3 bus1=A Pmpp=1000 kV=13.8 kVA=1010 effcurve=Eff
~ P-TCurve=FatorPvsT %Pmpp=100 daily=Irrad Tdaily=Temp  wattpriority=yes

New XYcurve.generic npts=5 yarray=[1 1 0 -1 -1] xarray=[0.5 0.92 1.0 1.08 1.5]

New InvControl.VoltVar mode=VOLTVAR vvc_curve1=generic RefReactivePower=VARMAX
```

Monitor mode 3 – State Variables

New Monitor.PV_v element=PVSystem.PV terminal=1 mode=3

Function Flags

hour	Irradiance	PanelkW	P_TFactor	Efficiency	Vreg	Vavg (DRC)	volt-var	volt-watt	DRC	VV_DRC
1	0	0	1	1	0.99	9999	0	9999	9999	9999
2	0	0	1	1	0.99	9999	0	9999	9999	9999
3	0	0	1	1	0.99	9999	0	9999	9999	9999
4	0	0	1	1	0.99	9999	0	9999	9999	9999
5	0	0	1	1	0.99	9999	0	9999	9999	9999
6	0	0	1	1	0.99	9999	0	9999	9999	9999
7	0.1	100	1	1	0.99	9999	0	9999	9999	9999
8	0.2	200	1	1	0.99	9999	0	9999	9999	9999
9	0.3	300	1	1	1.00	9999	-1	9999	9999	9999
10	0.5	500	1	1	1.01	9999	-1	9999	9999	9999

Regulated Voltage used in the Smart Inverter Functions

REQUEST Reactive Power

Volt-var, DRC, Watt-pf, Watt-var, and VV_DRC Flags

Negative value for var absorption

Positive value for var generation

1 indicates the function operates without restriction

0.6 indicates inverter kVA rating exceed under P priority

0.2 indicates reactive power limited by kvarMax/kvarMaxAbs

0 indicates no reactive power requested

LIMIT Active Power

Volt-watt Flag

1 indicates active power limited

0 otherwise

9999 indicates not applied

Volt-Var Example

```
New PVSystem.PV phases=3 Pmpp=1000 kV=13.8 kVA=1010 wattpriority=yes …
New XYcurve.generic npts=5 yarray=[1 1 0 -1 -1] xarray=[0.5 0.92 1.0 1.08 1.5]
New InvControl.VoltVar mode=VOLTVAR vvc_curve1=generic  RefReactivePower=VARMAX
```



Volt-Watt Example

```
New XYcurve.generic npts=3 yarray=[1 1 0] xarray=[1 1.02 1.1]
New InvControl.VoltWatt mode=VOLTWATT voltwatt_curve=generic VoltwattYAxis=PMPPPU
```



3. Storage + StorageController

Storage Element in OpenDSS

- General purpose energy storage
- In snapshot mode, same effect as a load/generator
- The strength is in time-varying (QSTS) simulation modes
  - Storage is a **variable** resource, but it is also **limited**
  - Account for **energy stored**, and **losses**

Modeling



Modeling



Operation – Power Flow within Storage

- In Charging state:



Operation – Power Flow within Storage

- In Discharging state:



Operation – Power Flow within Storage

- In Idling state:



Dispatch Modes

- Someone needs to tell the element when charge/discharge and at what rate
- "Self-dispatch":
  - Active power → *dispmode*
  - Reactive Power → *pf*, *kvar*
- Any combination of Active Power and Reactive Power control is valid!

Measure	Means	Modes
Active Power	Self-Dispatch	Default
		Follow
		LoadLevel
		Price
		External
		TimeChargeTrigger (Charge Only)
	Storage Controller	PeakShave/I-PeakShave (Discharge Only)
		Follow (Discharge Only)
		Support (Discharge Only)
		Schedule (Discharge Only)
		PeakShaveLow/I-PeakShaveLow (Charge Only)
		Loadshape
		Time
Reactive Power	Self-Dispatch	Constant PF
		Constant kvar
	InvControl	Volt-Var
		Dynamic Reactive Current (DRC)
		Volt-Var + DRC

Dispatch Modes – Self-Dispatch (Follow)




```
New LoadShape.dispatch_shape interval=1
npts=24 mult = [0, -1.0, -1.0, -1.0, -0.5, -0.5, 0, 0, 0,
0, 0, 0, 0 , 0, 0.5, 0.75, 1.0, 1.0, 1.0, 1.0, 0.75, 0.5, 0,
0]

New Storage.Storage1 phases=3 bus1=A kv=0.48

~ kWrated=50 %reserve=20

~ kWhrated= 500 %stored=50

~ dispmode=follow daily=dispatch_shape
```

$$P[t] = mult[t] * kWRated$$



Dispatch Modes – Self-Dispatch (External)

```
// 1am-2am
Set number=2
Solve
// 3am-7am
Edit Storage.Storage1 state=charging
~ %charge=80
Set number=5
Solve
// 8am-5pm
Edit Storage.Storage1 state=idling
Set number= 10
Solve
// 6am-10pm
Edit Storage.Storage1 kW=25
Set number=5
Solve
// 11pm-12am
Edit Storage.Storage1 state=idling
Set number= 2
Solve
```

Implicitly sets *state* to discharge (>0)




```
New Storage.Storage1 phases=3 bus1=A kv=0.48 kWrated=50 %reserve=20
kWhrated= 500 %stored=50 state=idling dispmode=External
```



Storage Controller Element in OpenDSS

- Control Element
- Designed to control a fleet of ES
- Monitors a terminal of a current-carrying element (typically a PDE)



Storage Controller - Dispatch Modes

Requested State	Modes
Discharging	Peakshave
	I-Peakshave
	Follow
	Support
	Schedule
Charging	PeakShaveLow
	I-PeakShaveLow
Discharging and Charging	Time
	Loadshape

Storage Controller - Dispatch Modes

- Examples – Base Circuit (IEEE 8500 Buses)



Fleet Energy Capacity= 7.35 MWh

Fleet Power Capacity= 1.55 MW

```
New Storage.A phases=3 bus1=l3235258   kv=12.47 kWhrated=500.0  kWrated=100.0  %stored=70
New Storage.B phases=3 bus1=m1069483 kv=12.47 kWhrated=1000.0 kWrated=200.0 %stored=70
New Storage.C phases=3 bus1=p862322   kv=12.47 kWhrated=1650.0 kWrated=350.0 %stored=70
New Storage.D phases=3 bus1=m1047615 kv=12.47 kWhrated=1250.0 kWrated=300.0 %stored=70
New Storage.E phases=3 bus1=m1069556 kv=12.47 kWhrated=500.0  kWrated=150.0 %stored=70
New Storage.F phases=3 bus1=l2688692   kv=12.47 kWhrated=1200.0 kWrated=200.0 %stored=70
New Storage.G phases=3 bus1=m1089131 kv=12.47 kWhrated=1250.0 kWrated=250.0 %stored=70
```

Storage Controller - Dispatch Modes – PeakShaving (Discharging) and Time (Charging)

```
New StorageController.SC element=Line.ln5815900-1 modedis=peakShave  kwtarget=10200
~ modecharge=Time timeChargeTrigger=2 %rateCharge=50
```





Storage Controller - Dispatch Modes – PeakShaving (Discharging) and Time (Charging)

```
New StorageController.SC element=Line.ln5815900-1 modedis=peakShave kwtarget=10200
~ modecharge=Time timeChargeTrigger=2 %rateCharge=50
```

No Power - Case

```
set casename=NoPower
Edit Storage.A kWrated=50
Edit Storage.B kWrated=100
Edit Storage.C kWrated=175
Edit Storage.D kWrated=150
Edit Storage.E kWrated=75
Edit Storage.F kWrated=100
Edit Storage.G kWrated=125
```



Original Fleet Power Capacity= 1.55 MW → **Fleet Power Capacity= 0.775 MW**

Storage Controller - Dispatch Modes – PeakShaving (Discharging) and Time (Charging)

```
New StorageController.SC element=Line.ln5815900-1 modedis=peakShave kwtarget=10200
~ modecharge=Time timeChargeTrigger=2 %rateCharge=50
```

No Energy - Case

```
set casename=NoEnergy
Edit Storage.A kWhrated=250
Edit Storage.B kWhrated=500
Edit Storage.C kWhrated=825
Edit Storage.D kWhrated=625
Edit Storage.E kWhrated=250
Edit Storage.F kWhrated=600
Edit Storage.G kWhrated=625
```



Original Fleet Energy Capacity= 7.35 MWh → **Fleet Energy Capacity= 3.675 MWh**

Storage Controller + InvControl




```
New Storage.B phases=3 bus1=B kv=13.8 kVA=1000 kWrated=1000 kWhrated= 10000 kvarMax=800
New Storage.C phases=3 bus1=C kv=13.8 kVA=1000 kWrated=1000 kWhrated= 10000 kvarMax=800
New Storage.D phases=3 bus1=D kv=13.8 kVA=1000 kWrated=1000 kWhrated= 10000 kvarMax=800
```

Storage Controller + InvControl




```
! StorageController
new Storagecontroller.SC element=line.AB
~ modedis=peakshave kwtarget=4000
```

```
! InvControl
New XYcurve.generic npts=5 yarray=[1 1 0 -1 -1]
~ xarray=[0.5 0.92 1.0 1.08 1.5]
New InvControl.InvControl mode=VOLTVAR
~ vvc_curve1=generic
~ RefReactivePower=VARMAX
```

```
!Cases
set casename=NoControls
Edit StorageController.SC enabled=False
Edit InvControl.InvControl enabled=False

set casename=Ppriority
Batchedit Storage..* wattpriority = True

set casename=Qpriority
Batchedit Storage..* wattpriority = False
```

```
set maxcontroliter=200
Set mode=Daily
Set stepsize=1h

Set number=10
Solve

Edit Vsource.source pu=0.94
set number=4
Solve

Edit Vsource.source pu=1
set number=10
Solve
```

Storage Controller + InvControl



New Monitor.Mon_StorageD_State element=Storage.D mode=3

Storage Controller + InvControl

- Monitor in Mode 3 for Storage D and Q-priority case

Dispatched Power

Requested by Storage Controller

hour	kWh	kWOut	kvarOut	Vref	kWDesired
11	8185.26	**600.086**	**799.936**	0.911377	**801.688**



Where do I go from here?

- Detailed technical notes available at your local OpenDSS installation Doc folder (*C:\Program Files\OpenDSS\Doc*)
  - *Inverter Modeling*
    - *DSS Scripts at "Examples/InverterTechNote/"*
  - *PVSystem and InvControl Element Models*
    - *DSS Scripts at "Examples/InverterModels/"*
  - *Storage Element*
    - *DSS Scripts at "Examples/StorageTechNote/"*
  - *StorageController Element*
    - *DSS Scripts at "Examples/StorageControllerTechNote/"*
- Official Forum at SourceForge



Together...Shaping the Future of Electricity